
04 PRÁCTICA FINAL
NO. 04

Contador binario con pantalla LCD

SESIÓN
4 de 4

DURACIÓN
4 horas

UNIDADES
7 y 8

PLATAFORMA
PICAXE‐14M2

Introducción

En la Práctica 2 el 74LS74 contaba con flip-flops: cada pulso de reloj sumaba uno y los LEDs mostraban
el número en binario. Hoy el mismo trabajo lo hace unmicrocontrolador con un programa. Cada vez que
se presiona el botón SUMAR la cuenta sube en uno; cuatro LEDs la muestran en binario y la pantalla LCD
la escribe en decimal y en binario. Un puente llamado BORRAR la regresa a cero, como la entrada CLR del
flip-flop.

El PICAXE‐14M2 es un microcontrolador de 14 terminales que se programa en BASIC. A diferencia del
Arduino, no viene en una placa: se coloca en el protoboard como cualquier integrado. Para cargarle el
programa se arma un pequeño circuito de programación con un 74LS04 , la misma compuerta NOT de la
Práctica 1.

El circuito se arma en tres etapas: primero el PICAXE y los LEDs, luego las entradas y al final la pantalla.
Cada etapa se prueba antes de pasar a la siguiente.

NOTA

Los grupos que trabajan con el Arduino Uno tienen su propia guía. El circuito hace exactamente lo
mismo; solo cambian el microcontrolador y la forma de cargar el programa.

OBJETIVOS

Armar el circuito mínimo de un microcontrolador: alimentación y programación.
Conectar salidas (LEDs) y entradas (botón y puente) al PICAXE.
Leer un número de 4 bits en binario y pasarlo a decimal.
Mostrar datos en una pantalla LCD por el bus I2C.
Cambiar el comportamiento del circuito modificando el programa.

MATERIALES Y EQUIPO

PICAXE‐14M2 y un 74LS04 .
Adaptador USB-serie ( CH340 o FTDI ) con cables jumper, y una computadora con el PICAXE
Editor y el controlador del adaptador.
MO‐MB102 con ADP‐5HB y el UT33B+ .
Pantalla LCD 16×2 con módulo I2C y cuatro cables jumper macho-hembra.
4 LEDs y 4 resistencias de 330Ω.
1 botón pulsador, 1 resistencia de 1 kΩ, 3 de 10 kΩ y 1 capacitor de 100 nF.
Protoboard y cables jumper.

PLANIFICACIÓN DE LA SESIÓN

Parte Actividad Tiempo

— Apertura: contar en binario con 4 bits 20 min

A El PICAXE, su programación y los LEDs 65 min

B Las entradas: SUMAR y BORRAR 35 min

C La pantalla LCD 40 min

D Retos: cambiar el programa 35 min

— Evaluación en Socrative, desarmado y cierre 25 min

Total 220 min
+ 20 min

Fundamento teórico

Contar con 4 bits

Cada LED es un bit. De derecha a izquierda valen 1, 2, 4 y 8: el número es la suma de los LEDs encendidos.
Con 4 bits se cuenta de 0 a 15. Al sumar uno a 15 (1111) el resultado necesitaría un quinto bit, así que la
cuenta regresa a 0, igual que un contador de flip-flops.

bit 3 bit 2 bit 1 bit 0 Valor

8 4 2 1 Peso de cada LED

0 1 1 0 4 + 2 = 6
1 0 1 1 8 + 2 + 1 = 11

Una entrada nunca queda al aire

El botón une el terminal con +5V. Cuando está suelto, la resistencia de 10 kΩ lo lleva a GND. Sin ella la
entrada queda flotando y la cuenta avanza sola, como pasó con las entradas TTL al aire en la Práctica 1.

+5V

al terminal

10 kΩ

Suelto: el terminal lee 0.
Presionado: el terminal lee 1. Por la resistencia pasan apenas
5V/10 kΩ = 0.5mA.
El rebote: al cerrarse, el contacto vibra unos milisegundos y la
entrada cambia varias veces. El programa espera 20ms y, ade-
más, no vuelve a contar hasta que se suelta el botón.

Cómo se programa el PICAXE

El PICAXE trae de fábrica un pequeño programa que escucha el terminal 2 (Serial In). Cuando la computado-
ra envía un programa nuevo, lo guarda y lo ejecuta; el chip lo conserva aunque se apague. Las respuestas
salen por el terminal 13 (Serial Out).

El PICAXE usa la señal serie invertida: en reposo la línea está en 0. Un adaptador USB-serie común la
entrega al revés, en reposo en 1. La solución es una compuerta 74LS04 en cada sentido:

Elemento Para qué está

NOT de los terminales 5 y 6 Invierte lo que manda la computadora (TXD) antes de llegar al terminal 2

NOT de los terminales 9 y 8 Invierte lo que responde el PICAXE (terminal 13) antes de llegar a RXD

1 kΩ en serie Protege el terminal 2

10 kΩ a GND Mantiene el terminal 2 en 0 aunque no haya nada conectado. Nunca se
quita: con el terminal 2 al aire el programa no corre

NOTA

En el manual del PICAXE la resistencia en serie es de 22 kΩ, porque allí la señal viene del cable oficial.
Aquí la entrega una salida TTL, cuyo nivel alto es de unos 3.4V: con 22 kΩ el terminal 2 recibiría
apenas 1V, y con 1 kΩ recibe 3.1V.

La pantalla por I2C

El módulo soldado detrás de la LCD recibe los datos por solo dos cables: SDA (datos) al terminal 9 y SCL

(reloj) al terminal 10. Su dirección suele ser 0x27; algunos módulos usan 0x3F. El PICAXE escribe la dirección
multiplicada por dos: $4E o $7E. Lo que se verá al final es esto:

Decimal: 06

Binario: 0110

Terminales

14M2

+V 1

C.5 Serial In 2

C.4 3

C.3 4

C.2 5

C.1 6

C.0 7

0V14

B.0 Serial Out13

B.112

B.211

B.3 SCL10

B.4 SDA9

B.58

Figura 04.1. PICAXE‐14M2 . Los terminales 11 y 12 no se
usan.

74LS04

1A 1

1Y 2

2A 3

2Y 4

3A 5

3Y 6

GND 7

VCC14

6A13

6Y12

5A11

5Y10

4A9

4Y8

Figura 04.2. 74LS04 . Se usan dos de sus seis
compuertas.

ADVERTENCIA

En el PICAXE la alimentación está en las esquinas de la izquierda: +V es el terminal 1 y 0V el 14. Es
al revés que en el 74LS04 , donde VCC es el 14 y GND el 7. Invertir la alimentación destruye el chip.

El circuito

PICAXE
14M2

Serial In
2

Serial Out
13

C.3
4

B.5
8

C.4
3

C.2
5

C.1
6

C.0
7

B.4 SDA
9

B.3 SCL
10

+V

1

0V

14

+5V

100nF

Ad
ap
ta
do
r

U
SB
‐s
er
ie

TXD

RXD

GND

1 kΩ

10 kΩ

5 6

98

74LS04

SUMAR

+5V

10 kΩ
BORRAR (puente)

+5V

10 kΩ

330Ω
bit 3

330Ω
bit 2

330Ω
bit 1

330Ω
bit 0

LCD 16×2
módulo I2C

SDA

SCL

VCC

GND

+5V

Figura 04.3. Esquema completo. A la izquierda, el circuito de programación; a la derecha, los LEDs y la LCD. Los
números pequeños son los terminales. El puente BORRAR es un cable que se pone y se quita.

Tabla 04.1. Conexiones, terminal por terminal

Terminal Se conecta a Parte

PICAXE‐14M2

1 (+V) Riel + A

14 (0V) Riel − A

2 (Serial In) 1 kΩ desde 04‐6, y 10 kΩ a GND A

13 (Serial Out) 04‐9 A

3, 5, 6, 7 Cada uno a una resistencia de 330Ω y su LED: el 3 (C.4) es el bit 3 y el 7
(C.0) el bit 0

A

4 (C.3) Botón SUMAR (el otro lado a +5V) y 10 kΩ a GND B

8 (B.5) Puente BORRAR (el otro lado a +5V) y 10 kΩ a GND B

9 (SDA) / 10 (SCL) SDA / SCL de la LCD C

11, 12 Libres —

74LS04

14 / 7 Riel + / riel − A

5 / 6 TXD del adaptador / 1 kΩ al terminal 2 del PICAXE A

9 / 8 Terminal 13 del PICAXE / RXD del adaptador A

Adaptador USB‐serie

GND Riel − A

5V o VCC No se conecta: el circuito se alimenta con el MO‐MB102 —

NOTA

En esta guía 04‐6 quiere decir “terminal 6 del 74LS04 ”.

El circuito armado

+ +
− −

− −
+ +

1

1

5

5

10

15

20

25

25

30

30

a a

b b

c c

d d

e e

f f

g g

h h

i i

j j

no se
conecta

02
(5
V
)

74LS04 PICAXE‐14M21 2

3

3

3

34

5

6

7

8

8

8

bit 3 bit 2 bit 1 bit 0

9

10

11

SUMAR BORRAR

12

13

14

15

16

17

18

19

Decimal: 06

Binario: 0110

GND
VCC
SDA
SCL

20 21

Figura 04.4. El circuito armado. Los números naranjas remiten a la lista de armado. Las letras (a a j) son las filas y
los números, las columnas. Los dos integrados tienen la muesca a la izquierda.

Tabla 04.2. Lista de armado

N.o Qué se coloca y dónde Parte

1 74LS04 en las columnas 3 a 9, muesca a la izquierda A

2 PICAXE‐14M2 en las columnas 13 a 19, muesca a la izquierda A

3 Alimentación, cuatro cables: rojo de a3 al riel + (04‐14); negro de j9 al riel − (04‐7); ne-
gro de a13 al riel − (0V del PICAXE); rojo de j13 al riel + (+V del PICAXE)

A

4 Capacitor de 100 nF entre los dos rieles de abajo, en la columna 11 A

5 Resistencia de 1 kΩ: de h8 a h14 (04‐6 al terminal 2 del PICAXE) A

6 Resistencia de 10 kΩ: de j14 al riel − A

7 Cable: de b14 a b8 (terminal 13 del PICAXE a 04‐9) A

8 Adaptador: GND al riel −, RXD a d9 (04‐8) y TXD a i7 (04‐5). Su pin de 5V no se conecta A

9 Cuatro resistencias de 330Ω cruzando la ranura: f22–e22, f24–e24, f26–e26 y f28–
e28

A

10 Cuatro LEDs: pata larga en a22, a24, a26 y a28; pata corta en el riel − de arriba A

11 Cables: j15 a j22 (bit 3), i17 a i24 (bit 2), h18 a h26 (bit 1) y g19 a g28 (bit 0) A

N.o Qué se coloca y dónde Parte

12 Botón SUMAR cruzando la ranura, con las patas en e10, e12, f10 y f12 B

13 Cable rojo: de a10 al riel + de arriba B

14 Resistencia de 10 kΩ: de j12 al riel − de abajo B

15 Cable: de g12 a g16 (al terminal 4 del PICAXE) B

16 Cable: de c19 a c30 (terminal 8 del PICAXE) B

17 Resistencia de 10 kΩ: de a30 al riel − de arriba B

18 Cable rojo: de j30 al riel + de abajo B

19 Puente BORRAR: un cable de d30 a g30. Se pone para borrar y se quita B

20 LCD: GND al riel − de arriba y VCC al riel + de arriba C

21 LCD: SDA a d18 (terminal 9) y SCL a d17 (terminal 10) C

RECOMENDACIÓN

El circuito de programación une los dos integrados dos veces, ambas entre las columnas 8 y 14: abajo
con la resistencia de 1 kΩ (lo que entra al PICAXE por el terminal 2) y arriba con el cable naranja (lo
que sale por el terminal 13). El botón usa dos patas en diagonal: e10 y f12; así funciona aunque se
coloque girado.

ADVERTENCIA

Apague el MO‐MB102 antes de mover cualquier cable. Coloque el módulo con sus dos puentes en
5V y verifique con el multímetro cuál riel es el +. Si su protoboard tiene los rieles partidos a la mitad,
una las dos mitades con un cable.

Procedimiento

Parte A. El PICAXE, su programación y los LEDs 65 min

1. Con la alimentación apagada, coloque los elementos 1 a 7 de la lista de armado. Todavía no conecte el
adaptador.

2. Encienda el MO‐MB102 y mida con el UT33B+ entre los terminales 1 y 14 del PICAXE, y entre el 14 y
el 7 del 74LS04 . Apague.

3. Coloque los LEDs (elementos 9 a 11) y conecte el adaptador (elemento 8). Conecte el adaptador a la
computadora.

4. Encienda el MO‐MB102 . En el PICAXE Editor abra PF_Contador_LCD.bas (su código completo está al
final de esta guía). La línea #picaxe 14M2 ya elige el chip; en Settings elija el puerto COM del adaptador
y pulse Program.

5. Al terminar la descarga, la luz debe caminar una vez del bit 0 al bit 3. Después los LEDs se apagan: la
cuenta está en 0. Apague y encienda el MO‐MB102 para repetir la prueba.

Tabla 04.3. Primera prueba

Comprobación Resultado

Tensión entre los terminales 1 y 14 del PICAXE

Tensión entre los terminales 14 y 7 del 74LS04

El programa se descargó sin errores (sí / no)

Orden en que encienden los LEDs al arrancar

Al terminar la prueba, ¿cuántos LEDs quedan en-
cendidos?

ADVERTENCIA

Mientras el adaptador esté conectado a la computadora, todo funciona. Si lo desconecta del USB, retire
también su cable TXD del protoboard: el adaptador apagado deja 04‐5 en 0, la NOT entrega un 1 al
terminal 2 y el PICAXE se queda esperando un programa. Con 04‐5 al aire, la entrada TTL se lee como
1 y el PICAXE corre normalmente.

Parte B. Las entradas: SUMAR y BORRAR 35 min

1. Apague y coloque los elementos 12 a 19. El puente BORRAR no se deja puesto.
2. Encienda. Presione SUMAR una vez por renglón de la tabla y anote los LEDs en el orden bit 3, bit 2, bit

1, bit 0 (encendido = 1).
3. Al final, ponga y quite el puente BORRAR.

Tabla 04.4. El contador en los LEDs

Pulsaciones desde 0 LEDs (bit 3 a bit 0) En decimal

1

2

3

4

7

8

10

15

16

Después de poner el puente
BORRAR

NOTA

Si la cuenta avanza sola o salta de dos en dos, revise la resistencia de 10 kΩ del elemento 14: sin ella
el terminal 4 queda al aire.

Parte C. La pantalla LCD 40 min

1. Apague y conecte la LCD (elementos 20 y 21).
2. Encienda. La pantalla muestra el saludo mientras la luz camina y después las dos líneas Decimal: y

Binario:.
3. Si la luz de fondo enciende pero no se ven letras, gire con un desarmador el potenciómetro azul del

módulo (contraste).
4. Lleve la cuenta a cada número de la tabla y anote lo que dice la LCD.

Tabla 04.5. La LCD y los LEDs

Lleve la cuenta a LCD: Decimal LCD: Binario ¿Coincide con los LEDs?

5

9

12

14

Tabla 04.6. El rebote

Prueba La LCD marca

Ponga el puente BORRAR, quítelo y presione SUMAR 10
veces, lo más rápido que pueda

Parte D. Retos: cambiar el programa 35 min

Cada reto cambia una o dos líneas. Guarde el programa con otro nombre antes de empezar.

1. Contador de década. Haga que la cuenta vaya de 0 a 9 y después regrese a 0, como el 74LS90 . Pista:
busque el número 15 en principal.

2. Contar hacia abajo. Haga que BORRAR reste uno en lugar de ir a 0, y que debajo de 0 siga el 15, como
el 74LS193 .

Si algo no funciona

Síntoma Revise

El editor no muestra el puerto
COM

El controlador del adaptador ( CH340 o FTDI ); otro cable USB

Hardware not found al descargar MO‐MB102 encendido; TXD a 04‐5 y RXD a 04‐8 (no al revés); GND del
adaptador al riel −; alimentación del 74LS04

El programa no corre La de 10 kΩ del terminal 2 (elemento 6); el cable TXD con el adaptador
desconectado del USB

El PICAXE se calienta Alimentación invertida: +V es el terminal 1. Apague de inmediato

Un LED nunca enciende Su polaridad y que la resistencia cruce la ranura en la misma colum-
na

La cuenta avanza sola La resistencia de 10 kΩ de j12 al riel −
SUMAR no hace nada El cable rojo de a10; el cable de g12 a g16

Los LEDs muestran otro número Cables de los terminales 3, 5, 6 y 7 cruzados: el 3 es el bit 3

LCD encendida, sin letras El contraste: potenciómetro azul del módulo

LCD con cuadros en la primera
línea

La dirección: cambie $4E por $7E y vuelva a descargar

LCD apagada VCC y GND del módulo; SDA (terminal 9) y SCL (terminal 10) cruzados

Preguntas de cierre

1. ¿Por qué después de 15 la cuenta regresa a 0?

2. ¿Para qué sirve el 74LS04 en el circuito de programación?

3. ¿Qué pasaría si se quita la resistencia de 10 kΩ del botón SUMAR?

4. ¿Cuántos cables necesita la LCD y qué lleva cada uno?

El programa completo

PF_CONTADOR_LCD.BAS

1 ' Universidad Rural de Guatemala - Electronica Digital FE731

2 ' Practica final, version PICAXE-14M2: contador binario con pantalla LCD

3 ' SUMAR suma 1 y BORRAR regresa a 0. La cuenta, de 0 a 15, se ve en

4 ' cuatro LEDs y en la LCD (I2C), en decimal y en binario.

5

6 #picaxe 14M2

7

8 symbol SUMAR = pinC.3 ' terminal 4, boton

9 symbol BORRAR = pinB.5 ' terminal 8, puente a +5 V

10 symbol LED0 = C.0 ' terminal 7, bit 0 (derecha)

11 symbol LED1 = C.1 ' terminal 6, bit 1

12 symbol LED2 = C.2 ' terminal 5, bit 2

13 symbol LED3 = C.4 ' terminal 3, bit 3 (izquierda)

14

15 ' El PICAXE usa la direccion de 8 bits: 0x27 por 2 = $4E.

16 ' Si la LCD solo muestra cuadros, use $7E (0x3F por 2).

17 symbol DIR_LCD = $4E

18

19 symbol cuenta = b0 ' sus bits se llaman bit0, bit1, bit2 y bit3

20 symbol dato = b1 ' byte que se envia a la LCD

21 symbol modo = b2 ' 0 = orden para la LCD, 1 = letra

22 symbol mitad = b3

23 symbol conEN = b4

24 symbol sinEN = b5

25 symbol k = b6

26

27 ' Textos de la pantalla, guardados en la EEPROM

28 eeprom 0, ("Decimal:") ' posiciones 0 a 7

29 eeprom 8, ("Binario:") ' posiciones 8 a 15

30 eeprom 16, ("URURAL FE731") ' posiciones 16 a 28

31 eeprom 29, ("Contador binario") ' posiciones 29 a 44

32

33 inicio:

34 hi2csetup i2cmaster, DIR_LCD, i2cslow, i2cbyte

35 gosub iniciarLCD

36

37 modo = 0 : dato = $80 : gosub enviarByte ' linea 1

38 modo = 1

39 for k = 16 to 28 : read k, dato : gosub enviarByte : next k

40 modo = 0 : dato = $C0 : gosub enviarByte ' linea 2

41 modo = 1

42 for k = 29 to 44 : read k, dato : gosub enviarByte : next k

43

44 ' Prueba de los LEDs: la luz camina del bit 0 al bit 3

45 for k = 0 to 3

46 lookup k, (1, 2, 4, 8), cuenta

47 gosub ponerLEDs

48 pause 400

49 next k

50

51 cuenta = 0

52 modo = 0 : dato = $01 : gosub enviarByte ' borrar la pantalla

53 pause 2

54 dato = $80 : gosub enviarByte

55 modo = 1

56 for k = 0 to 7 : read k, dato : gosub enviarByte : next k

57 modo = 0 : dato = $C0 : gosub enviarByte

58 modo = 1

59 for k = 8 to 15 : read k, dato : gosub enviarByte : next k

60 gosub mostrar

61

62 principal:

63 if SUMAR = 1 then

64 cuenta = cuenta + 1

65 if cuenta > 15 then

66 cuenta = 0

67 endif

68 gosub mostrar

69 pause 20 ' deja pasar el rebote

70 do while SUMAR = 1 : loop ' espera a que se suelte

71 pause 20

72 endif

73

74 if BORRAR = 1 then

75 cuenta = 0

76 gosub mostrar

77 pause 20

78 do while BORRAR = 1 : loop

79 pause 20

80 endif

81 goto principal

82

83 ponerLEDs:

84 if bit0 = 1 then : high LED0 : else : low LED0 : endif

85 if bit1 = 1 then : high LED1 : else : low LED1 : endif

86 if bit2 = 1 then : high LED2 : else : low LED2 : endif

87 if bit3 = 1 then : high LED3 : else : low LED3 : endif

88 return

89

90 mostrar:

91 gosub ponerLEDs

92 modo = 0 : dato = $89 : gosub enviarByte ' linea 1, columna 9

93 modo = 1

94 dato = cuenta / 10 + "0" : gosub enviarByte ' decenas

95 dato = cuenta // 10 + "0" : gosub enviarByte ' unidades

96 modo = 0 : dato = $C9 : gosub enviarByte ' linea 2, columna 9

97 modo = 1

98 dato = bit3 + "0" : gosub enviarByte

99 dato = bit2 + "0" : gosub enviarByte

100 dato = bit1 + "0" : gosub enviarByte

101 dato = bit0 + "0" : gosub enviarByte

102 return

103

104 ' ---- Manejo de la LCD por I2C (modulo PCF8574) ----

105 ' Cada byte del PCF8574 lleva: D7 D6 D5 D4 | luz | EN | RW | RS.

106 ' La LCD recibe cada byte en dos mitades de 4 bits; toma cada mitad

107 ' cuando EN pasa de 1 a 0.

108

109 enviarByte:

110 mitad = dato & $F0

111 gosub enviarMitad

112 mitad = dato * 16

113 gosub enviarMitad

114 return

115

116 enviarMitad:

117 conEN = mitad | %00001100 | modo

118 sinEN = mitad | %00001000 | modo

119 hi2cout (conEN, sinEN)

120 return

121

122 iniciarLCD:

123 pause 50

124 modo = 0

125 mitad = $30

126 gosub enviarMitad : pause 5

127 gosub enviarMitad : pause 1

128 gosub enviarMitad : pause 1

129 mitad = $20

130 gosub enviarMitad : pause 1 ' modo de 4 bits

131 dato = $28 : gosub enviarByte ' 2 lineas

132 dato = $0C : gosub enviarByte ' pantalla encendida, sin cursor

133 dato = $06 : gosub enviarByte ' el cursor avanza a la derecha

134 dato = $01 : gosub enviarByte ' borrar

135 pause 2

136 return

Parte del programa Qué hace

symbol Da nombre a los terminales y a las variables. cuenta es b0, y sus bits se leen
como bit0 a bit3

eeprom Guarda los textos de la pantalla en la memoria del chip

inicio Prepara la LCD, saluda y hace caminar la luz

principal Revisa SUMAR y BORRAR una y otra vez; espera el rebote y que se suelte el
botón

ponerLEDs y mostrar Encienden los LEDs con los bits de la cuenta y la escriben en la LCD

enviarByte y siguientes Mandan cada letra a la LCD en dos mitades de 4 bits por el bus I2C. No hace
falta modificarlas

Universidad Rural de Guatemala 14



ELECTRÓNICA DIGITAL • FE731
Manual de Laboratorio

Facultad de Ingeniería
Segundo Semestre 2026

EVALUACIÓN EN LÍNEA (SOCRATIVE)

Al terminar la práctica se aplicará una evaluación breve en Socrative. Las preguntas combinan la
teoría de la clase magistral en línea con lo que se acaba de hacer en el laboratorio.
El instructor indicará el código de la sala. Anótelo aquí, junto con cualquier indicación del día.
Código de sala / enlace

Espacio para la evaluación

Reservado para Socrative: el instructor colocará aquí el acceso o las indicaciones

